Problem:
0-1 Knapsack Problem

You have N items that you want to put them into a knapsack. Item i has value vi and weight wi.

You want to find a subset of items to put such that:

The total value of the items is as large as possible.

The items have combined weight at most W, that is capacity of the knapsack.

Find the maximum total value of items in the knapsack.

Input

N W
v1 w1
v2 w2
:
vN wN

The first line consists of the integers N and W. In the following lines, the value and weight of the i-th item are given.

Output

Print the maximum total values of the items in a line.

Constraints

1 ≤ N ≤ 100

1 ≤ vi ≤ 1000

1 ≤ wi ≤ 1000

1 ≤ W ≤ 10000

Sample Input 1

4 5
4 2
5 2
2 1
8 3

Sample Output 1

13

Sample Input 2

2 20
5 9
4 10

Sample Output 2

9

Solution in Haskell (judge verdict: Accepted)

{-# OPTIONS_GHC -O2 -funbox-strict-fields #-}
module Main(main) where
import qualified Data.Vector as V
import Data.Vector (Vector)
import Data.Vector ((!))
import Data.Vector ((!?))
import Control.Applicative
main :: IO ()
main = do
  n:maxw:_ <- map read . words <$> getLine :: IO [Int]
  vws <- V.replicateM n $ do
    v:w:_ <- map read . words <$> getLine :: IO [Int]
    return (v,w)
  print $ V.maximum $ solve vws (maxw+1)
  return ()
{-# INLINE solve #-}
solve :: Vector (Int,Int) -> Int -> Vector Int
solve vws maxw = V.foldl' next v0 vws
  where
    v0 :: Vector Int
    v0 = V.replicate (maxw) 0
    next :: Vector Int -> (Int,Int) -> Vector Int
    --next vs _ | traceShow vs False = undefined
    next vs (v,w) = V.generate maxw gen
      where
        gen i = max v1 v2
          where
            v1 = vs ! i
            v2 = case vs !? (i - w) of
              Nothing -> 0
              Just p -> p + v